



Generalising monads to arrows

John Hughes

Chalmers Tekniska Hogskola, Institutionen for Datavetenskap, S-412 96 Goteborg, Sweden

Abstract

Monads have become very popular for structuring functional programs since Wadler introduced their use in 1990. In particular, libraries of combinators are often based on a monadic type. Such libraries share (in part) a common interface, from which numerous benefits flow, such as the possibility to write generic code which works together with any library. But, several interesting and useful libraries are fundamentally incompatible with the monadic interface. In this paper I propose a generalisation of monads, which I call arrows, with significantly wider applicability. The paper shows how many of the techniques of monadic programming generalise to the new setting, and gives examples to show that the greater generality is useful. In particular, three non-monadic libraries for efficient parsing, building graphical user interfaces, and programming active web pages fit naturally into the new framework. © 2000 Elsevier Science B.V. All rights reserved.

1. Introduction

One of the distinguishing features of functional programming is the widespread use of *combinators* to construct programs. A combinator is a function which builds program fragments from program fragments; in a sense the programmer using combinators constructs much of the desired program automatically, rather than writing every detail by hand. The freedom that functional languages provide to manipulate functions – program fragments – as first-class citizens supports combinator programming directly.

Some combinators, such as the well-known list-processing operators *map* and *filter*, encapsulate generally useful program constructions and may appear in almost any functional program. Others are tailored to particular application areas, and are often collected into libraries that enable applications in that area to be built quickly and easily. For example, *parsing* is an application area that has been extensively studied.

E-mail address: rjmh@cs.chalmers.se (J. Hughes).

Given an appropriate library of parsing combinators, a parser for the grammar

$$G ::= a \ G \ b \mid c$$

might be programmed in Haskell [5,9] as

```
gram = symbol "a" 'cat' gram 'cat' symbol "b" <+> symbol "c"
```

A note on syntax: in Haskell, function application is written without brackets, so *symbol* "a" denotes a call of the function *symbol* with argument "a", and any function of two arguments may be used as an infix operator by enclosing it in back-quotes. In this example, *symbol* is a function which constructs a parser that accepts just the given token, 'cat' is a binary operator which combines two parsers into a parser that runs both in sequence (and concatenates their results), <+> is a binary operator which combines two parsers into one which tries both as alternatives, and the entire declaration is a recursive definition of a parser *gram* which recognises the language described by the grammar *G*.

Although the idea of programming with combinators is quite old, the design of combinator libraries has been profoundly influenced in recent years by Wadler's introduction of the concept of a *monad* into functional programming [13–15]. We shall discuss monads much more fully in the next section, but for now, suffice to say that a monad is a kind of standardised interface to an abstract data type of 'program fragments'. The monad interface has been found to be suitable for many combinator libraries, and is now extensively used. Numerous benefits flow from using a common interface: to take just one example, Haskell has been extended with special constructions to make the use of monads particularly convenient.

It is therefore a matter for some concern when libraries emerge which cannot, for fundamental reasons, use the monad interface. In particular, Swierstra and Duponcheel have developed a very interesting library for parsing LL-1 grammars [11], that avoids a well-known inefficiency in monadic parsing libraries by combining the construction of a parser with a 'static analysis' of the program so constructed. Yet Swierstra and Duponcheel's optimisation is incompatible with the monad interface. We believe that their library is not just an isolated example, but demonstrates a generally useful paradigm for combinator design that falls outside the world of monads. We shall look more closely at their idea in Section 3.

Inspired by Swierstra and Duponcheel's library, I sought a generalisation of the monad concept that could also offer a standardised interface to libraries of this new type. My proposal, which I call *arrows*, is the subject of this paper. Pleasingly, the arrow interface turned out to be applicable to other kinds of non-monadic library also, for example the *fudgets* library for graphical user interfaces [3], and a new library for programming active web pages. These applications will be described in Sections 6 and 10.

While arrows are a little less convenient to use than monads, they have significantly wider applicability. They can therefore be used to bring the benefits of monad-like programming to a much wider class of applications.

2. Background: Library design using monads

What, then, is a monad? In Haskell, the monad interface can be defined as a *class*:

```
class Monad m where
  return :: a → m a
  (≫=) :: m a → (a → m b) → m b
```

Read this as follows: a parameterised type m (which we may think of as a function from types to types) is a monad if it supports the two operations *return* and $\gg=$ (pronounced ‘bind’) with the types given. Intuitively, we think of a value of type $m\ a$ as representing a *computation* with result of type a – a program fragment. The nature of the computation is captured by the choice of the type m . The *return* operation constructs a trivial computation that just delivers its argument as its result. The $\gg=$ operation combines two computations in sequence, passing the result of the first as an argument to the second – hence the type of the second argument of $\gg=$: it is a function that constructs the second computation, rather than just a computation. The monad combinators must satisfy some simple laws, but we defer a discussion of these to Section 7.

2.1. An Example: A monad to manage failures

For example, computations which may raise an exception can be modelled in a purely functional language by returning a special value from functions which ‘fail’, and exception handling can be modelled by testing for this special value. We can use the parameterised type *Maybe* for this purpose, defined by

```
data Maybe a = Just a | Nothing
```

This declaration introduces a new parameterised type *Maybe* with two constructors, *Just* and *Nothing*. A value of type *Maybe a* is either of the form *Just x*, where x is a value of type a , or of the form *Nothing*. Now a function which intuitively returns a result of type t , but may fail, can be defined to return a result of type *Maybe t* instead, where *Nothing* represents failure. Other functions can ‘handle’ the exception by testing for *Nothing*, or propagate it by returning *Nothing* themselves. However, a disadvantage of this approach is that failure has to be propagated *explicitly*: at every function call, we have to test whether a failure occurred. The idea can be used much more conveniently if we define a combinator library to take care of failure handling.

To do so, we declare the type *Maybe* to be a monad; that is, we give implementations of *return* and $\gg=$ for this type. In Haskell, we write

```
instance Monad Maybe where
  return a = Just a
  x ≫= f  = case x of
    Just a  → f a
    Nothing → Nothing
```

where $x \gg= f$ fails immediately, without calling f , if its first argument x fails.

Using these combinators we can write functions which handle failure properly without any explicit tests for *Just* and *Nothing*. For example, the following function adds together two possibly failing integers, failing itself if either argument does:

```
add :: Maybe Int → Maybe Int → Maybe Int
add x y = x >>= λa →
           y >>= λb →
           return (a + b)
```

(The layout here is well suited to monadic programs, but may be confusing at first: the body of the λ -expression $\lambda a \rightarrow \dots$ extends to the *end* of the entire right-hand side!) Note that this definition is quite independent of the *Maybe* type: it only uses *return* and $\gg=$, and so would work with any monad.

To complete a useful library for failure handling we must add a combinator to cause a failure,

```
fail :: Maybe a
fail = Nothing
```

and a combinator to invoke a handler when a failure occurs:

```
handle :: Maybe a → Maybe a → Maybe a
x 'handle' h = case x of
                 Just a   → Just a
                 Nothing  → h
```

Now, we can treat the *Maybe* type as abstract, and write programs that cause and propagate failures just using the operators *fail*, *handle*, *return* and $\gg=$, without any explicit dependence on the way that failures are represented.

2.2. Another example: A monad to manage state

As another example, an updatable state can be modelled in a purely functional language by passing each function the current contents of the state as an additional parameter, and returning the possibly modified state as a part of each function's result. To do so by hand is tedious and error-prone, but fortunately we can encapsulate the state passing mechanism in a combinator library by using a monad.

In this case we represent a computation with result type *a* and a state of type *s* by a value of the type¹

```
newtype StateMonad s a = SM (s → (a,s))
```

¹ The Haskell **newtype** declaration introduces a new type isomorphic to an existing one, where the constructor names the isomorphism. Its purpose is to enable us to define overloaded operations which behave *differently* on the new type and the old. In this case, we will define the monad operations for *StateMonad* quite independently of any definition that may be given for functions. The difference between a **newtype** and **data** declaration is subtle, but quite important for efficiency: at run-time, a value of a type defined by **newtype** is represented in exactly the same way as the type it is isomorphic to – the constructor is not represented at all – whereas the constructors of a **data** type are always represented explicitly.

For any state type s , the partially applied type *StateMonad* s (which denotes a parameterised type with one remaining parameter) is a monad:

```
instance Monad (StateMonad s) where
  return a = SM (\s → (a,s))
  x >>= f  = SM (\s → let SM x' = x
                        (a,s') = x' s
                        SM f' = f a
                        in f' s')
```

With these definitions, we can write programs which pass around a state just in terms of *return* and *>>=*; there is no need to manipulate the state explicitly. Note that *>>=* must pass the modified state s' returned by its first argument to its second, rather than the original state, and must return the modified state returned by its second argument as part of its own result. If one attempts to pass a state around by hand, rather than by using combinators, then it is very easy to forget a $'$ somewhere, with strange bugs as a result.

To complete a library for state passing we must provide combinators for reading and modifying the state. For example,

```
fetch :: StateMonad s s
fetch = SM (\s → (s,s))

store :: s → StateMonad s ()
store x = SM (\s → ((),x))
```

Now the *StateMonad* type can be made abstract, and stateful programs can be written just in terms of the combinators. For example, a function to increment a counter:

```
tick :: StateMonad Int Int
tick = fetch >>= \n →
      store (n + 1) >>= \() →
      return n
```

This defines a computation with a state of type *Int*, which fetches the state and binds it to n , updates the state with $n + 1$, and then delivers the value of the counter n as its result.

2.3. Monadic parsing combinators

In practice, combinator libraries are usually based on monads providing a combination of features. For example, a parser for values of type a can be represented by the type

```
newtype Parser s a = P ([s] → Maybe (a,[s]))
```

where s is the type used to represent symbols in the parser's input, and $[s]$ is Haskell's notation for the type *list-of- s* . Such a parser is invoked by applying its representation to a list of symbols to parse; its result indicates whether or not parsing was successful, and in the event of success contains both the value parsed and the remaining, unparsed input. For example, a parser which recognises a particular symbol can be defined by

```
symbol :: s → Parser s s
symbol s = P (λxs → case xs of
                [] → Nothing
                (x : xs') → if x = s then Just (s, xs') else Nothing)
```

This parser fails if the input is empty or begins with the wrong symbol, and succeeds with one symbol consumed from the input otherwise.

This representation of parsers supports a combination of failure handling and state passing, where the state is the unparsed input. It can be declared to be a monad just like the *Maybe* and *StateMonad* types above – see Wadler's articles for details. Indeed, both the type and the implementation of the monad operators can be derived from the *Maybe* monad and a generalisation of our *StateMonad* – we will sketch this in Section 9. Further combinators can then be added to build up a complete library for parsing based on this monad.

This *Parser* type is often generalised to handle ambiguous grammars by replacing *Maybe* by a list, so that a parser may return several possible parses. We refrain from doing so here because Swierstra and Duponcheel's combinators, which we are most interested in, are designed to parse only *LL*(1) grammars.

2.4. Why use monads?

We have now seen that monads *can* be used as a basis for combinator libraries, but why *should* they be used? Why have monads become so ubiquitous in Haskell programs today?

One reason, of course, is that using monads simplifies code dramatically. It should be clear that writing a parser with explicit tests for failure and explicit passing of the input here and there, would be much more labour intensive than writing one in terms of *symbol*, *return* and \gg . However, this is an advantage of using *any* combinator library to encapsulate coding details, and does not argue for using monads in particular.

Another reason for using monads is that they offer a design guideline for combinator libraries: it is often a good start to begin by defining a suitable monad. For example, it is fairly clear that a library for parsing should include a combinator to invoke two parsers in sequence, but there are many possible ways in which such a combinator might handle the two parsers' results. In some early parsing libraries the two results were paired together, in others the sequencing combinator took an extra parameter, a function to combine the results. The monadic operator \gg is more general than either

of these: both may be easily defined in terms of $\gg=$, but the converse is not true. By basing a parsing library on a monad, the designer gives the user more flexibility than these ad hoc alternatives. Indeed, we know from experience that the monadic interface gives the library user great power.

Choosing a monadic interface can also guide the implementor of a combinator library. We have already seen three examples of monads; in fact, using monad transformers [6,7], we can systematically construct an infinite variety of monads. A systematic approach to monad design helps the implementor to find an appropriate type to base a combinator library on.

Finally, the fact that the monad operations *return* and $\gg=$ are overloaded in Haskell permits us to write *generic* monadic code, which can be used together with any library based on a monad. A growing collection of such functions are provided in the standard Haskell library. For example, we can generalise the *add* function above (for adding two possibly failing integers) into a generic function which applies *any* binary operator to the results of two computations.

$$\begin{aligned} \text{liftM2} &:: \text{Monad } m \Rightarrow (a \rightarrow b \rightarrow c) \rightarrow m\ a \rightarrow m\ b \rightarrow m\ c \\ \text{liftM2 } \text{op } x\ y &= x \gg= \lambda a \rightarrow \\ &\quad y \gg= \lambda b \rightarrow \\ &\quad \text{return}(a\ \text{'op'}\ b) \end{aligned}$$

(The $\text{Monad } m \Rightarrow$ in the type of *liftM2* is a *context*, and means that this function may be used for any monad type *m*.) Now the ‘*cat*’ operator on parsers that we saw in the introduction can be defined simply as

$$\text{cat} = \text{liftM2 } (++)$$

(where $++$ is Haskell’s concatenation operator for lists).

Generic code of this sort represents functionality that the designer of an individual combinator library no longer needs to provide: simply by basing the library on a monad, one gains access to a host of useful functions for free. This in turn may significantly reduce the work required to produce each new library.

Taken together, these arguments provide rather compelling reasons for using monads in combinator design; it is no wonder that they have become so ubiquitous.

2.5. Further parsing combinators

Let us pursue our example of combinators for parsing a little further. One of the things a parser can do is to fail; to enable us to express this we define a combinator which always fails. In fact, very many monads support a notion of failure, and so it is useful to overload the failure operator, just as we overloaded monadic *return* and

⇒. In Haskell this is done via a predefined class

```
class Monad m ⇒ MonadZero m where
    zero :: m a
```

to be read as follows: a parameterised type m is a *MonadZero* if it is a *Monad*, and additionally supports the operation *zero*. The implementation of *zero* for parsers is then defined by

```
instance MonadZero Parser where
    zero = P (λs → Nothing)
```

Moreover, many monads which support failure also support a *choice* combinator, which tries two alternative ways to perform a computation, using the second if the first one fails. Haskell defines a predefined class

```
class MonadZero m ⇒ MonadPlus m where
    (++) :: m a → m a → m a
```

and the implementation for parsers is

```
instance MonadPlus Parser where
    P a ++ P b = P (λs → case a s of
        Just (x, s') → Just (x, s')
        Nothing → b s)
```

This is the operator we called $\langle + \rangle$ in the example in the introduction.

This is one of the fundamental building blocks of a parsing library: every interesting grammar defines some non-terminals via alternatives. But unfortunately, this definition contains a serious *space leak*. That is, it can cause the retention of data by the garbage collector much longer than one would naively anticipate, with the result that parsers built with this operator may use much more space than one would reasonably expect. To see why, note that the input to be parsed, s , cannot be garbage collected while the first parser a is running, because if a eventually fails, then s must be passed to b . In a lazy language such as Haskell, it is the very act of running parser a which forces the list of tokens s to be constructed, perhaps by reading from a file. Provided a fails quickly, without forcing the evaluation of many elements of s , then little space is used. But if a actually succeeds in parsing a large part of the input s , then a great deal of space may be used to hold these already-parsed tokens, just in case a eventually fails and b needs to be invoked. Ironically, in practice a and b usually recognise quite different syntactic constructs, so that if a succeeds in parsing many symbols then b will almost certainly fail as soon as it is invoked. Saving the input for b is costly only when it is unnecessary!

This problem has been known since combinator libraries for parsing were first proposed, and Wadler for example gives a partial solution in his 1985 paper [12]. But the

solutions known for monadic parser libraries are only partial, and depend on the programmer using an additional combinator similar to Prolog's 'cut' operator, to declare that a parser need never backtrack beyond a certain point. Although monadic parser libraries work quite well in practice, the fundamental problem remains unsolved, which is really rather unsatisfactory.

3. Swierstra and Duponcheel's parsing library

In 1996, Swierstra and Duponcheel found a novel way to avoid the space leak in parsing libraries. They restrict their attention to LL(1) parsers, in which choices between alternative parses can always be resolved by looking at the next token of the input. Their implementation of $a++b$ can therefore choose between a and b immediately, and there is no need to save the input s in case the other alternative needs to be tried later. The space leak that other parsing libraries suffer from is completely cured.

To implement this idea, Swierstra and Duponcheel need to be able to tell, given a parser, which tokens it might accept as the first in the input, and also whether or not it can accept the empty sequence of tokens. (The latter information is needed to derive the former; a string matching the grammar PQ may begin with a token that starts Q , if P can match the empty string.) This means that parsers can no longer be represented as functions, as they were in the previous section. Instead, they are represented as a combination of static information, which can be computed before parsing begins, and a function to do the parsing itself. The combinators make use of the static information to construct 'optimised' parsing functions. Paraphrasing Swierstra and Duponcheel, we might define

```

data StaticParser s      = SP Bool [s]
newtype DynamicParser s a = DP ([s] → Maybe (a,[s]))
data Parser s a          = P (StaticParser s) (DynamicParser s a)

```

The first component of a parser tells us whether it matches the empty string, and which tokens it can accept first, while the second component is a function which does the actual parsing. For example, the combinator which accepts a particular symbol can be defined as

```

symbol :: s → Parser s s
symbol s = P (SP False [s]) (DP (λ(x:xs) → Just (s,xs)))

```

The dynamic parsing function need not test for an empty input, or check that the first symbol is s , because it will be invoked only when the preconditions expressed by the static part are satisfied.

Now we can make use of the static information to define the choice combinator efficiently:

instance *MonadPlus Parser where*

```

P (SP empty1 starters1) (DP p1) ++ P (SP empty2 starters2) (DP p2) =
  P (SP (empty1 ∨ empty2) (starters1 ++ starters2))
  (DP (λxs →
    case xs of
      []      → if empty1 then p1 [] else
                 if empty2 then p2 [] else Nothing
      x : xs' → if x ∈ starters1 then p1 (x : xs') else
                 if x ∈ starters2 then p2 (x : xs') else
                 if empty1 then p1 (x : xs') else
                 if empty2 then p2 (x : xs') else Nothing))

```

It is clear from this definition that the choice of whether to invoke p_1 or p_2 is made directly, and once made cannot be revised, so there is no need to retain a pointer to the input, and consequently no space leak.²

Just as the $++$ operator computes the starter symbols and potential emptiness of the parser it constructs, so must all of the other combinators. In most cases this is straightforward to do, but unfortunately in the case of $\gg=$ it turns out to be impossible! To see why, recall the type which $\gg=$ must have in this case:

$$(\gg=) :: \text{Parser } s \ a \rightarrow (a \rightarrow \text{Parser } s \ b) \rightarrow \text{Parser } s \ b$$

Now, the static properties of the result of $\gg=$ depend on the static properties of *both* the first and the second argument – for example, the combination can match the empty sequence only if both arguments can. Yet in the definition of $\gg=$, while we have access to the static properties of the first argument, we *cannot* obtain the static properties of the second one without applying it to a value of type a . Such values will be constructed only during parsing, but for Swierstra and Duponcheel’s idea to be useful we must compute the static parts of parsers once and for all, before parsing begins. It is simply impossible to find a definition of $\gg=$ which does this.

Swierstra and Duponcheel’s solution to this problem was to abandon the use of a monad: instead of $\gg=$ they defined a different sequencing operator with the type

$$(\diamond) :: \text{Parser } s \ (a \rightarrow b) \rightarrow \text{Parser } s \ a \rightarrow \text{Parser } s \ b$$

This operator is perfectly adequate for expressing parsers, and poses no problem as far as computing static properties in advance of parsing is concerned. Nevertheless, the

² However, this definition is not completely realistic. It assumes that the user of the library really does write an LL(1) parser, so that starters_1 and starters_2 are disjoint. In a real implementation this would of course be checked. Moreover, the expensive tests of the form $x \in \text{starters}_1$ can be avoided by choosing a cleverer representation of parsers – see Swierstra and Duponcheel’s article for details.

need to abandon the monad signature is worrying, for the reasons we discussed above. Useful as it is, Swierstra and Duponcheel’s parsing library stands alone; it cannot, for example, be used with generic monadic functions.

If this were an isolated case we might simply ignore it. But Swierstra and Duponcheel’s idea is clearly much more widely applicable: to optimise a combinator library, redefine the combinators to collect static properties of the computations they construct, and then use those static properties to optimise the dynamic computations. If we think of a library as defining a domain specific ‘language’, whose constructions are represented as combinators, then Swierstra and Duponcheel’s idea is to implement the language via a combination of a static analysis and an optimised dynamic semantics. We may clearly wish to do this very often indeed. But every time we do, the type of $\gg=$ will make it impossible to use a monadic interface!

It is this observation that motivated us to search for a generalisation of monads, a generic interface for combinator libraries that fits a much wider class of applications. We will introduce the generalisation we found in the next section.

4. Arrows

Recall that Swierstra and Duponcheel were unable to implement

$$(\gg=) :: \text{Parser } s \ a \rightarrow (a \rightarrow \text{Parser } s \ b) \rightarrow \text{Parser } s \ b$$

because its second argument is a function, and the only thing one can do with a function is apply it. Lacking a suitable value of type a to apply it to, they could not extract any static information from it, and therefore could not construct the static part of $\gg=$ ’s result.

Our solution is simply to *change the representation* of this argument. Rather than a function of type $a \rightarrow \text{Parser } s \ b$ we will use an abstract type, which we will call an *arrow* from a to b . We solve Swierstra and Duponcheel’s problem by choosing a representation for arrows which makes static properties immediately accessible.

In fact, there is no need to work with two abstract types, a monad type and an arrow type. Instead we will work purely with arrows. In general, an arrow type will be a parameterised type with *two* parameters, supporting operations analogous to *return* and $\gg=$. Just as we think of a monadic type $m \ a$ as representing a ‘computation delivering an a ’, so we think of an arrow type $a \ b \ c$ (that is, the application of the parameterised type a to the two parameters b and c) as representing a ‘computation with input of type b delivering a c ’.

Just as Haskell defines a *Monad* class, so we shall define an *Arrow* class with analogous operators. But we must make dependence on an input explicit. Thus while the *return* operator, with type $a \rightarrow m \ a$, merely converts a value into a computation, its analogue for arrows, with type $(b \rightarrow c) \rightarrow a \ b \ c$, converts a *function from input to output* into a computation. The analogue of $\gg=$ is just composition of arrows. We

define

```
class Arrow a where
  arr    :: (b → c) → a b c
  (≫) :: a b c → a c d → a b d
```

Category theorists might also expect an operator to construct the identity arrow from a type to itself; we omit it because it is directly obtainable as *arr* ($\lambda x \rightarrow x$).

For any monad *m*, functions of type $a \rightarrow m\ b$ are potential arrows. If we give this type a name,

```
newtype Kleisli m a b = K (a → m b)
```

then we can implement the arrow operations as follows:

```
instance Monad m ⇒ Arrow (Kleisli m) where
  arr f      = K (\b → return (f b))
  K f ≫ K g = K (\b → f b ≫ g)
```

This shows that arrows do indeed generalise monads; for every monad type, there is a corresponding arrow type. (Of course, it does not follow that every monadic *program* can be rewritten in terms of *arr* and \gg .) Categorically speaking, we just constructed the Kleisli category of the monad *m*. Of course, we will see later that there are also many other, non-monadic implementations of the arrow signature.

4.1. Arrows and pairs

However, even though in the case of monads the operators *return* and \gg are all we need to begin writing useful code, for arrows the analogous operators *arr* and \gg are not sufficient. Even the simple monadic addition function that we saw earlier

```
add :: Monad m ⇒ m Int → m Int → m Int
add x y = x ≫ λu → y ≫ λv → return (u + v)
```

cannot yet be expressed in an arrow form. Making dependence on an input explicit, we see that an analogous definition should take the form

```
add :: Arrow a ⇒ (a b Int) → (a b Int) → (a b Int)
add f g = ...
```

where we must combine *f* and *g* in sequence. The only sequencing operator available is \gg , but *f* and *g* do not have the right types to be composed. Indeed, the *add* function needs to *save the input* of type *b* across the computation of *f*, so as to be able to supply the same input to *g*. Likewise the result of *f* must be saved across the computation of *g*, so that the two results can eventually be added together and returned. The arrow combinators so far introduced give us no way to save a value across another computation, and so we have no alternative but to introduce another combinator.

We extend the definition of the *Arrow* class as follows:

```
class Arrow a where
  arr    :: (b → c) → a b c
  (>>>) :: a b c → a c d → a b d
  first  :: a b c → a (b, d) (c, d)
```

The new operator *first* converts an arrow from *b* to *c* into an arrow on pairs, that applies its argument to the first component and leaves the second component untouched, thus saving its value across a computation. Once again, we can implement *first* for any Kleisli arrow:

```
instance Monad m ⇒ Arrow (Kleisli m) where
  ...
  first (K f) = K (λ(b, d) → f b >>= λc → return(c, d))
```

Given *first*, we can define a combinator that applies its argument to the second component instead,

```
second :: Arrow a ⇒ a b c → a (d, b) (d, c)
second f = arr swap >>> first f >>> arr swap
where swap (x, y) = (y, x)
```

a combinator which processes both components of a pair,

```
(**) :: Arrow a ⇒ a b c → a d e → a (b, d) (c, e)
f ** g = first f >>> second g
```

and a combinator which builds a pair from the results of two arrows,

```
(&&&) :: Arrow a ⇒ a b c → a b d → a b (c, d)
f &&& g = arr (λb → (b, b)) >>> (f ** g)
```

With these definitions the *add* function is easily completed:

```
add :: Arrow a ⇒ (a b Int) → (a b Int) → (a b Int)
add f g = (f &&& g) >>> arr (λ(u, v) → u + v)
```

Just as we abstracted the idea of applying a binary operator to the results of two monadic computations, by going on to define *liftM2*, so we can generalise the arrow version likewise:

```
liftA2 :: Arrow a ⇒ (b → c → d) → a e b → a e c → a e d
liftA2 op f g = (f &&& g) >>> arr (λ(b, c) → b 'op' c)
```

We can then redefine *add* by

```
add = liftA2 (+)
```

Readers with a categorical background should note that Haskell pairs are *not* in general a product in the category of arrows. This would require properties such as

$$(f \&\& g) \ggg \text{arr fst} = f$$

to hold, and in general, since our arrows usually represent computations with some sort of effect, laws of this sort are simply false. In this case, the side-effects of g are lost on the right-hand side.

The reader may also wonder why we chose to take *first* as primitive, rather than (say) $\&\&$ which resembles a well-known categorical operator. There are two main reasons for our choice.

- Firstly, since in general our arrows represent computations with effects, evaluation order makes a difference. The definition of $f \&\& g$ above is explicit about this: the effects of f are composed with the effects of g *in that order*, that is evaluation is left to right. The definitions of $\&\&$ and $**$ above can be used as algebraic laws by the programmer, laws which capture evaluation order. In contrast, had we taken $\&\&$ as primitive, then the designer of each arrow-based library would have had to choose either left-to-right or right-to-left evaluation, with the result that evaluation order would probably differ from case to case. This would make the behaviour of arrow-based libraries less predictable, and reduce the number of useful laws that arrow combinators satisfy.
- Secondly, *first* is a simpler operator than $\&\&$, and in general its implementation is around half the size of that of the latter. In practice, the implementations of arrow combinators can be quite complex, and by making the choice we did we reduce the work required to build a new arrow-based library appreciably.

4.2. Swierstra and Duponcheel's parsers as arrows

Now let us return to Swierstra and Duponcheel's parsers. Recall that we defined their parser type as

```
data StaticParser s      = SP Bool [s]
newtype DynamicParser s a = DP ([s] → Maybe (a,[s]))
data Parser s a         = P (StaticParser s) (DynamicParser s a)
```

We were unable to make *Parser* into a monad, but can we make it into an arrow type?

To do so, we will need to add an extra type parameter, the type of inputs to parser arrows. Our intention is that the *static properties* of a parser should not depend on parse-time inputs, so let us change only the type of the dynamic parsing function:

```
newtype DynamicParser s a b = DP ((a,[s]) → Maybe (b,[s]))
data Parser s a b         = P (StaticParser s) (DynamicParser s a b)
```

Implementing the arrow combinators for this type is now straightforward:

instance *Arrow* (*Parser s*) **where**

arr f = *P (SP True []) (DP (λ(b,s) → Just (f b,s)))*

P (SP empty₁ starters₁) (DP p₁) \ggg *P (SP empty₂ starters₂) (DP p₂)* =
P (SP (empty₁ \wedge empty₂)
(starters₁ ‘union’ if empty₁ then starters₂ else []))
(DP (λ(b,s) → case p₁ (b,s) of
Just (c,s') → p₂ (c,s')
Nothing → Nothing

first (P sp (DP p)) =
P sp (λ((b,d),s) → case p (b,s) of
Just (c,s') → Just ((c,d),s')
Nothing → Nothing

It is easy to modify the definitions from Section 3 of *symbol*, the failure operator *zero*, and the choice combinator *++*, to handle the arrows' input appropriately. Of course, since *zero* and *++* are overloaded names for monad operators, then we cannot use the same names for the corresponding operators on arrows. We therefore introduce two further arrow classes,

class *Arrow a ⇒ ArrowZero a where*

zeroArrow :: *a b c*

class *ArrowZero a ⇒ ArrowPlus a where*

(<+>) :: *a b c → a b c → a b c*

and declare *Parser s* to be an instance of these classes instead. Having done so, we can go on to define all the operators in the interface that Swierstra and Duponcheel use, in terms of the arrow operations already introduced. For example, their sequencing operator is definable by

(⋈) :: *Parser s a (b → c) → Parser s a b → Parser s a c*
(⋈) = *liftA2 (λf x → f x)*

So the user of an arrow-based parsing library can use it in exactly the same way as Swierstra and Duponcheel's original library, but in addition can combine parsers with generic arrow code.

5. Arrows and interpreters

How awkward is it to program with arrow combinators instead of monadic ones? And how expressive are the combinators in each case – are there some kinds of program

which can be expressed using *return* and \gg , but cannot be written at all in terms of *arr*, \gg and *first*? We can begin to answer both questions by looking at (fragments of) an interpreter based on arrows *vs.* one based on monads. If we can write an interpreter in which program fragments in a certain language are interpreted as arrows, then we know that any kind of program expressible in the interpreted language can also be expressed in terms of the arrow combinators. Since we not concerned to implement any particular kind of effect, our interpreters will in both cases be generic, that is they will work with *any* monad or arrow, respectively.

To begin with, we shall consider a tiny language with only variables and addition. We represent expressions by the datatype

```
data Exp = Var Id | Add Exp Exp
type Id = String
```

The value of such an expression is always an integer, but in anticipation of making extensions we introduce a separate type of values anyway:

```
data Val = Num Int
```

We will also require a type for environments:

```
type Env = [(Id, Val)]
```

Now, a monadic interpreter maps expressions to computations, represented using a monad M . To do so, we introduce an evaluation function

$$eval :: Exp \rightarrow Env \rightarrow M\ Val$$

which we can define by

```
eval (Var s) env      = return (lookup s env)
eval (Add e1 e2) env = liftM2 add (eval e1 env) (eval e2 env)
                        where add (Num u) (Num v) = Num (u + v)
```

An arrow interpreter, on the other hand, maps expressions to computations represented as arrows. But what should the input of an arrow denoting an expression be? By analogy with the monadic case, it is natural to take the input of an expression to be the environment. In an arrow interpreter based on arrow type A , we therefore give *eval* the type

$$eval :: Exp \rightarrow A\ Env\ Val$$

We can define *eval* as follows:

```
eval (Var s)      = arr (lookup s)
eval (Add e1 e2) = liftA2 add (eval e1) (eval e2)
                        where add (Num u) (Num v) = Num (u + v)
```

As we can see, at least in this small example, the arrow code is by no means more awkward than the monadic code. Indeed, often the user of a monadic combinator

library works more with derived operators such as *liftM2* than with the operators in the monad signature themselves. Where analogous operators can be defined on arrows, arrow programs are essentially the same as monadic ones.

5.1. Interpreting conditionals

Let us pursue the interpreter example a little further, and add a conditional expression to the interpreted language. We extend the expression and value types as follows:

```
data Exp = ... | If Exp Exp Exp
data Val = ... | Bl Bool
```

The monadic interpreter is easy to extend; we add a new case

$$\text{eval } (\text{If } e_1 \ e_2 \ e_3) \ \text{env} = \text{eval } e_1 \ \text{env} \gg= \lambda(\text{Bl } b) \rightarrow \\ \text{if } b \text{ then } \text{eval } e_2 \ \text{env} \text{ else } \text{eval } e_3 \ \text{env}$$

But the arrow interpreter is more difficult. Certainly, we could define

$$\text{eval } (\text{If } e_1 \ e_2 \ e_3) = (\text{eval } e_1 \ \&\& \ \text{eval } e_2 \ \&\& \ \text{eval } e_3) \gg= \\ \text{arr } (\lambda(\text{Bl } b, (v_1, v_2)) \rightarrow \text{if } b \text{ then } v_1 \text{ else } v_2)$$

but this does not properly capture the meaning of a conditional expression: *both* branches are evaluated, and we just choose between the results.³ Of course the intention is to evaluate just *one* branch, depending on the value of the boolean.

And this is the crux of the problem: the arrow combinators provide no way to choose between two arrows on the basis of an input. To do so, we are obliged to add a new combinator. But this time, we choose to define a new class *ArrowChoice* rather than enlarge the existing *Arrow* class further. By doing so we retain the freedom to define arrow types which *do not* support a dynamic choice combinator; they will simply fail to be instances of our new class.

The new combinator we want will choose between two arrows on the basis of the input, and it makes sense therefore for the input to be of Haskell's sum type

```
data Either a b = Left a | Right b
```

We will define $(f ||| g)$ to pass *Left* inputs to f and *Right* inputs to g , so the type of $|||$ will be

$$(|||) :: \text{ArrowChoice } a \Rightarrow a \ b \ d \rightarrow a \ c \ d \rightarrow a \ (\text{Either } b \ c) \ d$$

However, just as we chose to define *first* as an arrow primitive rather than $\&\&$, so we choose a simpler operator than $|||$ as the primitive method in the *ArrowChoice*

³ Do not expect Haskell's lazy evaluation to avoid computing *eval* of the other branch. Normally the definition of arrow composition, and hence also $\&\&$, forces the result to depend on both operands, so lazy evaluation plays no rôle.

class. We define

```
class Arrow  $a \Rightarrow$  ArrowChoice  $a$  where
  left ::  $a \ b \ c \rightarrow a$  (Either  $b \ d$ ) (Either  $c \ d$ )
```

where *left f* invokes *f* only on *Left* inputs, and leaves *Right* inputs unchanged. As usual, we check that we can implement *left* for Kleisli arrows:

```
instance Monad  $m \Rightarrow$  ArrowChoice (Kleisli  $m$ ) where
  left (K  $f$ ) = K ( $\lambda x \rightarrow$  case  $x$  of
    Left  $b \rightarrow f \ b \gg= \lambda c \rightarrow$  return (Left  $c$ )
    Right  $d \rightarrow$  return (Right  $d$ ))
```

Once we have introduced *left*, we can define

```
right  $f =$  arr mirror  $\gg=$  left  $f \gg=$  arr mirror
where mirror (Left  $x$ ) = Right  $x$ 
        mirror (Right  $x$ ) = Left  $x$ 
```

```
 $f \# g =$  left  $f \gg=$  right  $g$ 
```

```
 $f ||| g =$  ( $f \# g$ )  $\gg=$  arr untag
where untag (Left  $x$ ) =  $x$ 
        untag (Right  $y$ ) =  $y$ 
```

Now returning to our interpreter, we can at last define the interpretation of conditionals:

```
eval (If  $e_1 \ e_2 \ e_3$ ) = (eval  $e_1 \ \&\& \text{arr id}$ )  $\gg=$ 
  arr( $\lambda(Bl \ b, env) \rightarrow$  if  $b$  then Left  $env$  else Right  $env$ )  $\gg=$ 
  (eval  $e_2 |||$  eval  $e_3$ )
```

This is certainly more awkward than the monadic code, but would be much simplified by introducing a combinator especially for testing predicates:

```
test :: Arrow  $a \Rightarrow a \ b \ Bool \rightarrow a \ b$  (Either  $b \ b$ )
test  $f =$  ( $f \ \&\& \text{arr id}$ )  $\gg=$  arr ( $\lambda(b, x) \rightarrow$  if  $b$  then Left  $x$  else Right  $x$ )
```

Such a combinator is sufficiently useful that it is reasonable to include it in the arrow library, whereupon this case of our interpreter becomes no more complicated than the monadic version:

```
eval (If  $e_1 \ e_2 \ e_3$ ) = test (eval  $e_1 \ \gg=$  arr( $\lambda(Bl \ b) \rightarrow b$ ))  $\gg=$  (eval  $e_2 |||$  eval  $e_3$ )
```

Alternatively, Paterson's syntactic sugar for arrows (see Section 8) makes this definition much more readable.

5.2. Interpreting λ -calculus

Using the combinators we have now introduced, we could go on to write an arrow interpreter for a complete first-order functional language. But can we interpret higher-order functions? Let us consider adding λ -expressions and (call-by-value) application to the interpreted language. We extend the type of expressions as follows:

data $Exp = \dots \mid Lam\ Id\ Exp \mid App\ Exp\ Exp$

Before we can extend the type Val , we must decide how to represent function values. Since calling a function may have an effect, we cannot interpret functions as values of type $Val \rightarrow Val$. In the monadic interpreter, we can use functions whose result is a computation,

data $Val = \dots \mid Fun\ (Val \rightarrow M\ Val)$

while in the arrow interpreter, we naturally represent functions by arrows:

data $Val = \dots \mid Fun\ (A\ Val\ Val)$

The monadic *eval* function is easily extended to handle the new cases:

$$\begin{aligned} eval\ (Lam\ x\ e)\ env &= return\ (Fun\ (\lambda v \rightarrow eval\ e\ ((x, v) : env))) \\ eval\ (App\ e_1\ e_2)\ env &= eval\ e_1\ env \gg \lambda f \rightarrow eval\ e_2\ env \gg \lambda v \rightarrow f\ v \end{aligned}$$

But the arrow version proves more difficult. Interpreting λ -expressions is unproblematic,

$$eval\ (Lam\ x\ e) = arr\ (\lambda env \rightarrow Fun\ (arr\ (\lambda v \rightarrow (x, v) : env) \gg eval\ e))$$

but application is harder. If we try to define

$$eval\ (App\ e_1\ e_2) = ((eval\ e_1 \gg arr\ (\lambda (Fun\ f) \rightarrow f)) \&\& eval\ e_2) \gg app$$

for some suitable definition of *app*, then we find that *app* must invoke an arrow *which it receives as an input*, and there is no way to do so using the combinators so far introduced. There is nothing for it but to introduce another new class:

class *Arrow* $a \Rightarrow ArrowApply\ a$ **where**
 $app :: a\ (a\ b\ c, b)\ c$

whereupon the definition of *eval* above works. So, given an implementation of *app*, we can write an interpreter for the λ -calculus, and so we can also express other arrow programs in a higher-order style.

Note that we can define *left* in terms of *app*:

$$\begin{aligned} left\ f &= arr\ (\lambda z \rightarrow \text{case } z \text{ of} \\ &\quad Left\ b \rightarrow (arr\ (\lambda () \rightarrow b) \gg f \gg arr\ Left, ()) \\ &\quad Right\ d \rightarrow (arr\ (\lambda () \rightarrow Right\ d), ())) \gg \\ &\quad app \end{aligned}$$

so any arrow type in class *ArrowApply* can also be made an instance of *ArrowChoice*.

Once more, it is easy to implement *app* for Kleisli arrows:

```
instance Monad m  $\Rightarrow$  ArrowApply (Kleisli m) where
  app = K ( $\lambda(K\ f, x) \rightarrow f\ x$ )
```

We have now seen that, given a monad *m*, we can define a corresponding arrow type *Kleisli m* which moreover supports all the other combinators we have introduced so far. Conversely, it turns out that, given an arrow type *a* which also supports *app*, we can define a corresponding monad type *ArrowMonad a*. The definition is simply

```
newtype ArrowApply a  $\Rightarrow$  ArrowMonad a b = M (a Void b)
```

where *Void* is Haskell's one-point type, whose only element is the undefined value \perp . That is, a 'monadic' computation based on *a* is simply an arrow which ignores its input. We can now define the monad operations on *ArrowMonad a*:

```
instance ArrowApply a  $\Rightarrow$  Monad (ArrowMonad a) where
  return x      = M (arr ( $\lambda z \rightarrow x$ ))
  M m  $\gg=$  f = M (m  $\gg=$ 
                    arr ( $\lambda x \rightarrow$  let M h = f x in (h,  $\perp$ ))  $\gg=$ 
                    app)
```

We need *app* in order to invoke the arrow that the second argument of $\gg=$ produces.

One conclusion we can draw from this is that arrow types which support *app* are just as expressive as monads. In principle, one might eliminate the concept of a monad from Haskell altogether, and replace it with arrows supporting *app*. But another conclusion to draw is that arrows supporting *app* are really of little interest to us here. Our motivation, after all, is to find a generic interface for combinator libraries which *cannot* be based on a monad. But clearly, any library which supports an arrow type with *app* could equally well be given a monadic interface. In the rest of the paper, therefore, we will be most interested in arrow types which *cannot* be made instances of *ArrowApply*.

What of Swierstra and Duponcheel's parsers? A moment's thought shows that parsers *cannot* support either choice or application. The choice operator $f \parallel g$ is supposed to make a dynamic choice between two arrows on the basis of the input, which implies that the possible starting symbols of $f \parallel g$ would depend on the arrow's input. But we have deliberately designed the *Parser* type so that the value of the input *cannot* affect the static component. It follows that \parallel is unimplementable. And if choice is unimplementable, then so a fortiori is *app*, since the former can be defined in terms of the latter. Luckily this does not matter: it is rare that we *want* to write a parser which decides on the grammar to accept on the basis of previously parsed values.

What we see here is that the arrow interface lets the programmer make finer distinctions than the monad interface does; we can distinguish between types of computations that permit dynamic choices and calls of dynamic functions, and types of computations that do not. Swierstra and Duponcheel parsers do not. In contrast, once we declare a type to be a monad, we open the possibility of doing everything with it. This is why the monadic interface is restrictive: the only kinds of computations we can represent

as monads are those which support *all* the operations of choice, application, and so on. But computations which support only some of these can still be represented as arrows, they just will not be instances of all the arrow classes.

It may seem surprising that we focus our attention on arrows without *app*, when Cartesian Closed Categories (which model the λ -calculus) otherwise play such a large rôle in computer science, or indeed, when higher-order programming is so prevalent in Haskell. But we are actually giving up very little, since when we use arrows we work with two ‘levels’ of function. Of course, we can still work with functions on functions, and even functions on arrows; the one thing we cannot express is arrows on arrows. It is a small price to pay for the generality which arrows give us.

6. Stream processors: Processes as arrows

We have already seen that any monad gives rise to a corresponding arrow type in a natural way, and that Swierstra and Duponcheel’s parsers (or more generally, combinators which collect static information about computations) can also be represented as arrows. In this section we will show that yet another ‘non-monadic’ notion of computation, namely that of a process, fits naturally into the arrow framework.

We concern ourselves for the time being with processes that have one input channel and one output channel. Such processes can be modelled in a purely functional language by *stream processors*. A stream processor maps a stream of input messages into a stream of output messages, but is represented by an abstract data type. Let $SP\ a\ b$ be the type of stream processors with inputs of type a and outputs of type b . Stream processors are then constructed using the operators

$$put :: b \rightarrow SP\ a\ b \rightarrow SP\ a\ b$$

which constructs a stream processor which outputs the b and then behaves like the second argument, and

$$get :: (a \rightarrow SP\ a\ b) \rightarrow SP\ a\ b$$

which constructs a stream processor which waits for an input, passes it to its function argument, and then behaves like the result. For simplicity, we shall only consider non-terminating (recursively defined) stream processors; otherwise we would add another operator to construct a stream processor which halts.

Stream processors can be represented in several different ways, but quite a good choice is as a datatype with *put* and *get* as constructors:

```
data SP a b = Put b (SP a b) | Get (a → SP a b)
put = Put
get = Get
```

Now, we can write single processes using *put* and *get*, but to put processes together we need further combinators.

The arrow combinators turn out to represent very natural operations on processes! The *arr* operator maps its argument *f* over the input stream, and the \ggg operator connects two processes in series:

instance Arrow SP where

$$\text{arr } f = \text{Get } (\lambda x \rightarrow \text{Put } (f \ x) (\text{arr } f))$$

$$sp_1 \ggg \text{Put } c \ sp_2 = \text{Put } c \ (sp_1 \ggg sp_2)$$

$$\text{Put } b \ sp_1 \ggg \text{Get } f = sp_1 \ggg f \ b$$

$$\text{Get } f_1 \ggg \text{Get } f_2 = \text{Get } (\lambda a \rightarrow f_1 \ a \ggg \text{Get } f_2)$$

⋮

Note that we define process composition *lazily*: the composition blocks waiting for an input only if *both* its constituent processes do.

Finally, the *first* operator builds a process that feeds the first components of its inputs through its argument process, while the second components bypass the argument process and are recombined with its outputs. But what if the argument process does not produce one output per input? Our solution is to buffer the unconsumed inputs until corresponding outputs are produced. The function *bypass* below takes as additional arguments the queues of unused first and second components; note that one of these is always empty.

instance Arrow SP where

⋮

$$\text{first } f = \text{bypass } [] [] f$$

$$\text{bypass } [] \ ds \ (\text{Get } f) = \text{Get } (\lambda(b,d) \rightarrow \text{bypass } [] \ (ds++[d]) \ (f \ b))$$

$$\text{bypass } (b : bs) [] \ (\text{Get } f) = \text{bypass } bs [] \ (f \ b)$$

$$\begin{aligned} \text{bypass } bs [] \ (\text{Put } c \ sp) &= \text{Get } (\lambda(b,d) \rightarrow \\ &\quad \text{Put } (c,d) \ (\text{bypass } (bs++[b]) [] \ sp)) \end{aligned}$$

$$\text{bypass } [] \ (d : ds) \ (\text{Put } c \ sp) = \text{Put } (c,d) \ (\text{bypass } [] \ ds \ sp)$$

With this definition, $f \&\& g$ combines *f* and *g* in parallel, synchronising their output streams to produce a stream of pairs (and also synchronising their joint output with the input stream).

We can now use generic arrow combinators to write down stream processors. For example, the following stream processor outputs Fibonacci numbers:

$$\text{fibs} = \text{put } 0 \ \text{fibs}'$$

$$\text{fibs}' = \text{put } 1 \ (\text{liftA2 } (+) \ \text{fibs} \ \text{fibs}')$$

Stream processors also support a natural notion of failure: a failing process simply never produces more output. We can therefore define a *zeroArrow* as

instance ArrowZero SP where

$$\text{zeroArrow} = \text{Get } (\lambda x \rightarrow \text{zeroArrow})$$

This is often called a sink. (Had we included a constructor for a stopped process in the type SP then we would of course use that here, rather than representing *zeroArrow* by the process which continually reads input.)

We define $p <+> q$ to run p and q in parallel, merging their outputs; this is often called a fanout.

instance ArrowPlus SP where

$$\begin{aligned} \text{Put } b \text{ } sp_1 <+> sp_2 &= \text{Put } b \text{ } (sp_1 <+> sp_2) \\ sp_1 <+> \text{Put } b \text{ } sp_2 &= \text{Put } b \text{ } (sp_1 <+> sp_2) \\ \text{Get } f_1 <+> \text{Get } f_2 &= \text{Get } (\lambda a \rightarrow f_1 a <+> f_2 a) \end{aligned}$$

We take care to define parallel composition lazily also, so that $p <+> q$ blocks waiting for input only if *both* p and q do.

These definitions satisfy the laws

$$\begin{aligned} \text{zeroArrow} <+> q &= q \\ p <+> \text{zeroArrow} &= p \\ (p <+> q) <+> r &= p <+> (q <+> r) \end{aligned}$$

which is a strong indication that they are reasonable.

Stream processors can also support dynamic choice. The stream processor *left sp* simply passes messages tagged *Left* through sp , while others are passed on directly.

instance ArrowChoice SP where

$$\begin{aligned} \text{left } (\text{Put } c \text{ } sp) &= \text{Put } (\text{Left } c) \text{ } (\text{left } sp) \\ \text{left } (\text{Get } f) &= \text{Get } (\lambda z \rightarrow \text{case } z \text{ of} \\ &\quad \text{Left } a \rightarrow \text{left } (f a) \\ &\quad \text{Right } b \rightarrow \text{Put } (\text{Right } b) \\ &\quad \quad (\text{left } (\text{Get } f))) \end{aligned}$$

With this definition, then $f ||| g$ can be regarded as yet another kind of parallel composition, which routes inputs tagged *Left* to f and inputs tagged *Right* to g .

In fact, although stream processors have only one input and one output channel, we can model processes with many of each by *multiplexing* several channels onto one. For example, we can regard a channel carrying messages of type *Either a b* as a representation for *two* channels, one carrying a 's and the other carrying b 's. With this viewpoint, $f ||| g$ combines f and g in parallel to yield a stream processor with two input channels (multiplexed onto one), and merges the output channels onto one. Should we wish to combine f and g without merging their outputs, we can instead use $f \# g$. We can copy an input channel to two output channels using $\text{arr Left} <+> \text{arr Right}$, and so we can define a parallel combination of f and g with two output channels, but which copies one input channel to both processes by

$$f \mid \& \mid g = (\text{arr Left} <+> \text{arr Right}) \ggg (f \# g)$$

We can write a stream processor with two input channels and one output, that just copies the first input channel and discards the second, or vice versa, as

$$\begin{aligned} \text{justLeft} &= \text{arr id} \parallel \text{zeroArrow} \\ \text{justRight} &= \text{zeroArrow} \parallel \text{arr id} \end{aligned}$$

Not surprisingly, combining two processes and then discarding the output channel from one of them is equivalent to the other:

$$\begin{aligned} (f \mid \& \mid g) \gg \text{justLeft} &= f \\ (f \mid \& \mid g) \gg \text{justRight} &= g \end{aligned}$$

But these properties have a categorical interpretation: they tell us that the *Either* type is a *weak categorical product* in the category of stream processors! (Only weak, because there is more than one way to define $\mid \& \mid$ so that these equations hold; our definition favours g over f in the case that both produce outputs simultaneously.) In a categorical sense, then, the *Either* type behaves more like a product than the pair type does, when we work with stream processors. And indeed, a channel carrying a sum type corresponds much more closely to a pair of channels than does a channel carrying pairs.

The only arrow class we have not yet shown how to implement is *ArrowApply*. But it turns out that there is no sensible definition of

$$\text{app} :: SP (SP\ a\ b, a)\ b$$

Since *app* would receive a new stream processor to invoke with every input, there is no real sense in which the stream processors it is passed would receive a *stream* of inputs; we could supply them with only one input each. This would really be very unnatural. Since stream processors do not support a natural definition of *app*, they cannot be fitted into the monadic framework. They thus give us our second example of a useful kind of computation which cannot be represented as a monad.⁴

However, recalling that *Either* may play the rôle of a product type for stream processors, we might instead of *app* consider looking for a function of type

$$\text{dyn} :: SP (Either (SP\ a\ b)\ a)\ b$$

There is actually a very natural definition with this type: the ‘dynamic stream processor’ *dyn* receives stream processors on its first input channel, and then passes inputs from its second input channel through the stream processor received, until it receives another

⁴ Another way to see this is to note that the Kleisli arrow of any monad M , with type $a \rightarrow M\ b$, takes only a *single* a as input. In contrast, stream processors take *many* a s as inputs, and so cannot be constructed as a Kleisli arrow.

stream processor to replace the first. We implement it as

```

dyn = dynloop zeroArrow
  where dynloop (Put b sp) = Put b (dynloop sp)
        dynloop (Get f) = Get (λz →
          case z of
            Right a → dynloop (f a)
            Left sp → dynloop sp)

```

Stream processors are not just amusing toys: they are at the heart of the *fudgets* combinator library for programming graphical user interfaces [3]. A fidget with the type $F\ a\ b$ can be thought of as a process with four communication channels, a *high-level input channel*, carrying messages of type a from other fidgets, a *high-level output channel* carrying messages of type b to other fidgets, and two *low-level channels* carrying messages to and from the window manager. A fidget can therefore exchange high-level messages with other fidgets, but can also manage a part of the screen. Thus a fidget has both an appearance and a behaviour, which makes them useful for structuring complex user interfaces.

The fidget type $F\ a\ b$ is actually implemented as a stream processor in which the high and low level communication channels are multiplexed onto one, in just the way we described. Since fidgets are just stream processors, they can also be declared to be arrows, supporting the same operations. Interestingly, almost all the operations we discussed in this section do indeed appear in the fidgets library – even *dyn* – although of course, they appear with different names, and not as instances of a general framework. If declarations making fidgets into arrows were added to the fidgets library, then programmers would be able to combine fidgets using generic arrow combinators, in addition to those which are already defined in the library.

7. Arrow laws

Up to this point we have ignored the matter of laws. In fact, the presentation of monads in Section 2 was a little oversimplified: an implementation of *return* and \gg constitutes a monad only if the so-called *monad laws* are satisfied:

$$\begin{aligned}
 \text{return } x \gg f &= f\ x \\
 m \gg \text{return} &= m \\
 (m \gg \lambda x \rightarrow f\ x) \gg g &= m \gg (\lambda x \rightarrow f\ x \gg g)
 \end{aligned}$$

These laws state in essence that sequential composition is associative, and *return* is its unit, although they are complicated slightly by the need to pass values from one computation to the next. The programmer relies implicitly on the monad laws every time he or she uses a monad-based library without worrying about how to bracket sequential compositions.

We will place similar requirements on the implementations of the arrow combinators. But since there are many more arrow combinators than monadic ones, we will require a larger collection of laws. All of the laws that we state in this section are satisfied by Kleisli arrows.

We can simplify the statements of the laws a little by noting that the ordinary function type can be declared to be an arrow:

instance *Arrow* (\rightarrow) **where**

$arr\ f = f$

$f \ggg g = g \circ f$

$first\ f = \lambda(b,c) \rightarrow (f\ b, c)$

instance *ArrowChoice* (\rightarrow) **where**

$left\ f\ (Left\ b) = Left\ (f\ b)$

$left\ f\ (Right\ d) = Right\ d$

instance *ArrowApply* (\rightarrow) **where**

$app = \lambda(f,x) \rightarrow f\ x$

Of course, we will require composition to be associative, and moreover to be preserved by *arr*:

$(f \ggg g) \ggg h = f \ggg (g \ggg h)$

$arr\ (f \ggg g) = arr\ f \ggg arr\ g$

Moreover, we require that *arr id* be an identity for composition:

$arr\ id \ggg f = f = f \ggg arr\ id$

Categorically speaking, we now know that arrows form a category, and that *arr* is a functor from the category of Haskell functions to the category of arrows.

These laws correspond in some sense to the monad laws, but now we must go on to state the laws that the other arrow combinators are required to satisfy. Let us call an arrow *pure* if it is equal to *arr f* for some *f*; a pure arrow ‘has no side-effects’. We shall require that all combinators behave for pure arrows as they do for functions; that is

$first\ (arr\ f) = arr\ (first\ f)$

$left\ (arr\ f) = arr\ (left\ f)$

Furthermore, we require that our combinators preserve composition:

$first\ (f \ggg g) = first\ f \ggg first\ g$

$left\ (f \ggg g) = left\ f \ggg left\ g$

Categorically, we require that *first* and *left* be functors. Similar properties for *second* and *right* follow as easy consequences.

Note, though, that it does *not* follow that

$$(f \mathbin{**} g) \ggg (h \mathbin{**} k) = (f \ggg h) \mathbin{**} (g \ggg k)$$

since the order of g and h differs on the two sides. This is another reason to favour *first* and *left* as primitives over their more usual binary counterparts: the laws they must satisfy become much simpler to state.

We formalise the property that *first* f depends only on first components of pairs as follows:

$$\text{first } f \ggg \text{arr fst} = \text{arr fst} \ggg f$$

but it is *not* in general true that

$$\text{first } f \ggg \text{arr snd} = \text{arr snd}$$

since, on the right-hand side, the side-effects of f are lost. Instead, we formalise the intuition that the second component of a pair is unaffected by *first* f via two laws.

The first of these laws allows a function of that second component to be moved across the use of *first*. We have to require that the function be pure, to avoid potentially changing the order in which side-effects occur. Thus the law becomes

$$\text{first } f \ggg \text{second } (\text{arr } g) = \text{second } (\text{arr } g) \ggg \text{first } f$$

Once again, the dual statement, in which *first* and *second* are interchanged, follows as an easy corollary.

The second law formalises the intuition that if *first* f saves the value of the second component across a call of f , then saving two values should be equivalent to saving a pair. Let

$$\text{assoc } ((x, y), z) = (x, (y, z))$$

We can think of *assoc* as moving y into the ‘saved state’. Now, saving y and then passing x to f should have the same effect as passing (x, y) to *first* f and then saving y . Formally,

$$\text{arr assoc} \ggg \text{first } f = \text{first } (\text{first } f) \ggg \text{arr assoc}$$

Given these laws we can prove that

$$((f \mathbin{**} g) \mathbin{**} h) \ggg \text{assoc} = \text{assoc} \ggg (f \mathbin{**} (g \mathbin{**} h))$$

We note in passing that many categorical properties of products fail in the presence of side effects. For example, the reader might expect that

$$f \ggg (g \mathbin{\&\&} h) = (f \ggg g) \mathbin{\&\&} (f \ggg h)$$

but this is not true (unless f is pure) because the side-effects of f are duplicated on the right.

The laws for *first* serve as models for the laws for *left*; we require that

$$\begin{aligned} \text{arr Left} \ggg \text{left } f &= f \ggg \text{arr Left} \\ \text{right (arr } g) \ggg \text{left } f &= \text{left } f \ggg \text{right (arr } g) \\ \text{arr assocsum} \ggg \text{left } f &= \text{left (left } f) \ggg \text{arr assocsum} \end{aligned}$$

where

$$\begin{aligned} \text{assocsum (Left (Left } x))} &= \text{Left } x \\ \text{assocsum (Left (Right } y))} &= \text{Right (Left } y) \\ \text{assocsum (Right } z) &= \text{Right (Right } z) \end{aligned}$$

Note here also that we cannot change the order of *left* *f* and *right* *g* unless we know that one of *f* or *g* is pure, because we might change the order of side-effects.

For arrows supporting application, we require firstly that ‘currying’ and then applying the identity arrow is equivalent to the identity (on pairs):

$$\text{first (arr } (\lambda x \rightarrow \text{arr } (\lambda y \rightarrow (x, y)))) \ggg \text{app} = \text{arr id}$$

Secondly, we require a kind of parametricity property for *app*, which permits operations to be moved in or out of the applied arrow:

$$\begin{aligned} \text{first (arr (} g \ggg \text{))} \ggg \text{app} &= \text{second } g \ggg \text{app} \\ \text{first (arr (} \ggg h)) \ggg \text{app} &= \text{app} \ggg h \end{aligned}$$

From these laws we can prove an analogue of η -conversion, that applying a constant arrow using *app* is equivalent to the arrow itself:

$$\text{arr } (\lambda x \rightarrow (f, x)) \ggg \text{app} = f$$

Moreover, currying and then applying any arrow is equivalent to the arrow:

$$\text{first (arr } (\lambda x \rightarrow \text{arr } (\lambda y \rightarrow (x, y)) \ggg f)) \ggg \text{app} = f$$

Finally, we can prove that the monad laws hold for the *ArrowMonad* defined in Section 5.2.

For the remaining arrow classes, *ArrowZero* and *ArrowPlus*, we just require that $\langle + \rangle$ is associative, and *zeroArrow* is its unit. Stronger conditions, such as, for example,

$$\text{zeroArrow} \ggg f = \text{zeroArrow}$$

would be overly restrictive: this property fails for stream processors, for example, since *f* may very well produce outputs independently of its input.

In general, there is something of a conflict between the desire on the one hand to state many laws, thus making it possible to prove strong properties generically, for every kind of arrow, and the wish on the other hand to leave open the possibility of very many different implementations of the arrow signature. We believe that the laws we have stated in this section are a rather minimal set, which every reasonable arrow type should satisfy.

8. Syntactic sugar for arrows

While monads can be used without any kind of language support, monadic programming in Haskell is made significantly more attractive by using syntactic sugar (the **do** notation). The notational simplicity which results is especially important for beginners. In the case of arrows, there is even more to be gained from wisely chosen syntax, since much of the ‘plumbing’ to carry values from one point to another can then be made implicit.

Ross Paterson has recently proposed an extension to Haskell to support arrows. We shall explain a minor variant of his proposal here, and show how much it can improve readability. We will make heavy use of the extension in the remaining sections of this paper.

Paterson proposes to add a new form of expression to Haskell, which we write as $\lambda \bullet p \rightarrow \{b\}$, denoting an arrow. Here p is a pattern, and the body of the arrow, b , may take one of the forms below (where e denotes an expression):

$$\begin{aligned}
 b &::= e \\
 &\quad | \quad e \bullet e \\
 &\quad | \quad p \leftarrow b; \quad b \\
 &\quad | \quad \text{if } e \text{ then } b \text{ else } b \\
 &\quad | \quad \text{case } e \text{ of } \{brs\} \\
 brs &::= p \rightarrow b \\
 &\quad | \quad p \rightarrow b; \quad brs
 \end{aligned}$$

As usual, curly brackets and semicolons can be omitted using Haskell’s layout rule.

Intuitively, a body of the form e just returns the value of e , an ‘arrow application’ $e_1 \bullet e_2$ passes the value of e_2 to the arrow e_1 , a binding $p \leftarrow b_1; \quad b_2$ binds the output from b_1 to the pattern p within b_2 , a conditional body **if** e **then** b_1 **else** b_2 chooses between two bodies based on the boolean value of e , and a case body **case** e **of** $\{brs\}$ chooses a branch whose pattern matches e .

The meaning of the new syntax is defined formally by the following translation:

$$\llbracket \lambda \bullet p \rightarrow \{b\} \rrbracket = \llbracket b \rrbracket_p$$

$$\begin{aligned}
\llbracket e \rrbracket_p &= \text{arr } (\lambda p \rightarrow e) \\
\llbracket e_1 \bullet e_2 \rrbracket_p &= \begin{cases} \text{arr } (\lambda p \rightarrow (e_1, e_2)) \ggg \text{app} \\ \text{arr } (\lambda p \rightarrow e_2) \ggg e_1, & \text{if } FV(e_1) \cap FV(p) = \emptyset \end{cases} \\
\llbracket p' \leftarrow b_1; b_2 \rrbracket_p &= \text{arr } (\lambda x \rightarrow (x, x)) \ggg \text{first } \llbracket b_1 \rrbracket_p \ggg \llbracket b_2 \rrbracket_{(p', p)} \\
\llbracket \text{if } e \text{ then } b_1 \text{ else } b_2 \rrbracket_p &= \text{arr } (\lambda p \rightarrow \text{if } e \text{ then Left } p \text{ else Right } p) \ggg \\
&\quad (\llbracket b_1 \rrbracket_p ||| \llbracket b_2 \rrbracket_p) \\
\llbracket \text{case } e \text{ of } \{brs\} \rrbracket_p &= \text{arr } (\lambda p \rightarrow (e, p)) \ggg \llbracket brs \rrbracket_p
\end{aligned}$$

$$\begin{aligned}
\llbracket p' \rightarrow b \rrbracket_p &= \llbracket b \rrbracket_{(p', p)} \\
\llbracket p' \rightarrow b; brs \rrbracket_p &= \text{arr } (\lambda(x, p) \rightarrow \text{case } x \text{ of} \\
&\quad p' \rightarrow \text{Left } (x, p) \\
&\quad z \rightarrow \text{Right } (x, p)) \ggg \\
&\quad (\llbracket b \rrbracket_{(p', p)} ||| \llbracket brs \rrbracket_p), \quad x \text{ fresh}
\end{aligned}$$

Arrow expressions are overloaded, and can be used to denote an arrow of any type. However, an arrow expression involving an **if** or a **case** must of course denote an arrow in the class *ArrowChoice*, since the translation of these constructions uses combinators from that class. Similarly, an arrow expression whose translation involves the combinator *app* must denote an arrow in the class *ArrowApply*.

Note, however, that there are *two* possible translations of ‘arrow application’ $e_1 \bullet e_2$. The first is always applicable, but involves *app*, and so requires the arrow expression to have a type in *ArrowApply*. The second translation uses only the basic arrow combinators, but is applicable only when the arrow e_1 does not involve any variables in the pattern p . Let us call these *arrow-bound variables*; they are just the variables bound by $\lambda \bullet p \rightarrow \dots$ or $p \leftarrow \dots$; An arrow expression thus requires its type to be in class *ArrowApply* if any expression to the left of an arrow application (\bullet) involves an arrow-bound variable.

Just as we usually define functions by one or more equations, rather than using λ -expressions and **case** expressions explicitly, we will allow ourselves to define arrows by equations in a similar way. We write

$$\begin{aligned}
f \bullet p_1 &= b_1 \\
&\vdots \\
f \bullet p_n &= b_n
\end{aligned}$$

instead of

$$\begin{aligned}
f = \lambda \bullet x \rightarrow \text{case } x \text{ of} \\
&\quad p_1 \rightarrow b_1 \\
&\quad \vdots \\
&\quad p_n \rightarrow b_n
\end{aligned}$$

We leave the details of this notation a little informal.

As an illustration of Paterson's notation, we rewrite the interpreter from Section 5:

$$\begin{aligned}
 \text{eval } (\text{Var } s) \bullet \text{env} &= \text{lookup } s \text{ env} \\
 \text{eval } (\text{Add } e_1 \ e_2) \bullet \text{env} &= \text{liftA2 } \text{add} \ (\text{eval } e_1) \ (\text{eval } e_2) \bullet \text{env} \\
 \text{eval } (\text{If } e_1 \ e_2 \ e_3) \bullet \text{env} &= \text{Bl } b \leftarrow \text{eval } e_1 \bullet \text{env} \\
 &\quad \text{if } b \text{ then eval } e_2 \bullet \text{env} \text{ else eval } e_3 \bullet \text{env} \\
 \text{eval } (\text{Lam } x \ e) \bullet \text{env} &= \text{Fun } (\lambda \bullet v \rightarrow \text{eval } e \bullet (x, v) : \text{env}) \\
 \text{eval } (\text{App } e_1 \ e_2) \bullet \text{env} &= \text{Fun } f \leftarrow \text{eval } e_1 \bullet \text{env} \\
 &\quad v \leftarrow \text{eval } e_2 \bullet \text{env} \\
 &\quad f \bullet v
 \end{aligned}$$

Comparison with the unsugared version should convince the reader of the value of Paterson's idea.

9. Functors: New arrows from old

One of the attractive features of monads is that they can be built up piecewise, using so-called *monad transformers* [7]. A monad transformer is a monad parameterised on another monad, such that computations over the parameter monad can be 'lifted' to computations over the new one.

For example, the state monad of Section 2.2 can be generalised to a monad transformer:

newtype *StateMonadT* *s m a* = *SM* (*s* → *m* (*a*, *s*))

In general, the monad operators on the new type must be defined in terms of the monad operators on the parameter monad, as in this case:

instance *Monad m* ⇒ *Monad (StateMonadT s m)* **where**

$$\begin{aligned}
 \text{return } a &= \text{SM } (\lambda s \rightarrow \text{return } (a, s)) \\
 x \gg= f &= \text{SM } (\lambda s \rightarrow \text{let SM } x' = x \text{ in} \\
 &\quad x' s \gg= \lambda(a, s') \rightarrow \\
 &\quad \text{let SM } f' = f \ a \text{ in} \\
 &\quad f' s')
 \end{aligned}$$

Lifting of computations is defined by passing the state through unchanged:

liftState :: *Monad m* ⇒ *m a* → *StateMonadT s m a*
liftState *x* = *SM* ($\lambda s \rightarrow x \gg= \lambda a \rightarrow \text{return } (a, s)$)

Finally, the new monad supports fetch and store operations, just like the original state monad:

fetch :: *Monad m* ⇒ *StateMonad s m s*
fetch = *SM* ($\lambda s \rightarrow \text{return } (s, s)$)

store :: *Monad m* ⇒ *s* → *StateMonad s m ()*
store *x* = *SM* ($\lambda s \rightarrow \text{return } ((), x)$)

The new monad thus supports all the computations of the parameter monad (by lifting), and in addition manages a state. By composing monad transformers together, one can build up a monad providing any desired combination of features. For example, if we want a monad which manages a state and handles failures, we can use the type *StateMonadT s Maybe*, and this is of course just the parsing monad we discussed in Section 2.3.

In this section we show that arrows have the same property: we can define ‘arrow transformers’ which map simpler arrow types to more complex ones. The most important monad transformers have arrow transformer counterparts, and we will describe those for handling failures, state, and continuations. An arrow transformer is, by analogy with a monad transformer, just an arrow type parameterised on another arrow type, such that arrows of the second type can be mapped into arrows of the first. It may be helpful to look at the *kind* of an arrow transformer: since an arrow type has kind $* \rightarrow * \rightarrow *$ (it maps two types into a type), then an arrow transformer must have the kind

$$(* \rightarrow * \rightarrow *) \rightarrow (* \rightarrow * \rightarrow *)$$

It maps an arrow type into an arrow type. If F is an arrow transformer with this kind, then the function mapping arrows of type a to arrows of type $F\ a$ must have the type

$$\text{lift}F :: a\ b\ c \rightarrow F\ a\ b\ c$$

We shall require arrow transformers to satisfy certain laws. We require that

$$\text{lift}F\ (\text{arr}\ f) = \text{arr}\ f$$

$$\text{lift}F\ (f \ggg g) = \text{lift}F\ f \ggg \text{lift}F\ g$$

$$\text{lift}F\ (\text{first}\ f) = \text{first}\ (\text{lift}F\ f)$$

where the arrow combinators on the left-hand sides are for arrows of type a , and those on the right are for arrows of type $F\ a$. These requirements imply that our arrow transformers are *functors* in the usual categorical sense, and so from now on we shall use the word functor instead of arrow transformer.

We note briefly that the concepts of monad transformers and functors can be formalised as classes, thus overloading the lifting operations, but that this requires a more powerful class system than Haskell currently supports. We therefore refrain from doing so.

9.1. The maybe functor

Any arrow type can be lifted to an arrow type supporting failures by the functor

$$\text{newtype MaybeFunctor } a\ b\ c = MF\ (a\ b\ (\text{Maybe}\ c))$$

That is, we use arrows whose result can indicate failure. We can lift arrows to this type using

$$\begin{aligned} \text{liftMaybe} &:: \text{Arrow } a \Rightarrow a\ b\ c \rightarrow \text{MaybeFunctor } a\ b\ c \\ \text{liftMaybe } f &= MF\ (f \ggg \text{arr Just}) \end{aligned}$$

The arrow operations need to handle failures, which means they need to make dynamic decisions. We therefore must require that the parameter arrow type supports choice:

```
instance ArrowChoice a  $\Rightarrow$  Arrow (MaybeFunctor a) where
  arr f                = liftMaybe (arr f)
  MF f  $\ggg$  MF g = MF ( $\lambda \bullet x \rightarrow z \leftarrow f \bullet x$ 
                     case z of
                       Just c  $\rightarrow g \bullet c$ 
                       Nothing  $\rightarrow$  Nothing)
  first (MF f)         = MF ( $\lambda \bullet (b, d) \rightarrow c' \leftarrow f \bullet b$ 
                             case c' of
                               Just c  $\rightarrow$  Just (c, d)
                               Nothing  $\rightarrow$  Nothing)
```

Arrows formed by *MaybeFunctor* support failure and failure handling, of course:

```
instance ArrowChoice a  $\Rightarrow$  ArrowZero (MaybeFunctor a) where
  zeroArrow = MF ( $\lambda \bullet z \rightarrow$  Nothing)

instance ArrowChoice a  $\Rightarrow$  ArrowPlus (MaybeFunctor a) where
  MF f <+> MF g = MF ( $\lambda \bullet b \rightarrow c' \leftarrow f \bullet b$ 
                     case c' of
                       Just c  $\rightarrow$  Just c
                       Nothing  $\rightarrow g \bullet b$ )
```

and they also, not surprisingly, support choice:

```
instance ArrowChoice a  $\Rightarrow$  ArrowChoice (MaybeFunctor a) where
  left (MF f) = MF (left f)
```

Finally, if the underlying arrows support application, then so do the arrows produced by *MaybeFunctor*:

```
instance (ArrowChoice a, ArrowApply a)  $\Rightarrow$ 
  ArrowApply (MaybeFunctor a) where
  app = MF ( $\lambda \bullet (MF f, b) \rightarrow f \bullet b$ )
```

9.2. The state functor

Any arrow type can be lifted to an arrow type supporting state passing by the functor

```
newtype StateFunctor s a b c = SF (a (b, s) (c, s))
```

We can lift arrows to this type using

```
liftState :: Arrow a  $\Rightarrow$  a b c  $\rightarrow$  StateFunctor s a b c
liftState f = SF (first f)
```

The arrow operations just pass the state along as one would expect:

```
instance Arrow a  $\Rightarrow$  Arrow (StateFunctor s a) where
  arr f          = liftState (arr f)
  SF f  $\ggg$  SF g = SF (f  $\ggg$  g)
  first (SF f)   = SF ( $\lambda \bullet ((b,d),s) \rightarrow (c,s') \leftarrow f \bullet (b,s)$ 
                       $((c,d),s')$ )
```

Of course, the arrows produced by the *StateFunctor* support fetch and store operations:

```
fetch :: Arrow a  $\Rightarrow$  StateFunctor s a b s
fetch = SF ( $\lambda \bullet (b,s) \rightarrow (s,s)$ )
```

```
store :: Arrow a  $\Rightarrow$  StateFunctor s a s ()
store = SF ( $\lambda \bullet (x,s) \rightarrow ((),x)$ )
```

Stateful arrows inherit the ability to support dynamic choice, failure, and failure handling from the parameter arrow:

```
instance ArrowChoice a  $\Rightarrow$  ArrowChoice (StateFunctor s a) where
  left (SF f) = SF ( $\lambda \bullet (z,s) \rightarrow$  case z of
    Left b   $\rightarrow (d,s') \leftarrow f \bullet (b,s)$ 
    (Left d,s')
    Right c  $\rightarrow (Right c,s)$ )
```

```
instance ArrowZero a  $\Rightarrow$  ArrowZero (StateFunctor s a) where
  zeroArrow = SF zeroArrow
```

```
instance ArrowPlus a  $\Rightarrow$  ArrowPlus (StateFunctor s a) where
  SF f <+> SF g = SF (f <+> g)
```

Finally, if the underlying arrow type supports application, then so do stateful arrows based on it:

```
instance ArrowApply a  $\Rightarrow$  ArrowApply (StateFunctor s a) where
  app = SF ( $\lambda \bullet ((SF f,b),s) \rightarrow f \bullet (b,s)$ )
```

The state functor we have defined is, of course, closely related to the state monad transformer, but the advantage of defining functors on arrows, rather than transformers on monads, is that we can apply them to arrow types that do not correspond to any monad. As an example, the reader is invited to work out the behaviour of arrows of type *StateFunctor s SP*, derived by adding state passing to stream processors.

9.3. The CPS functor

A third well-known monad transformer adds continuation passing to any monad. In the monadic world, we can define

```
newtype CPS ans m a = CPS ((a  $\rightarrow$  m ans)  $\rightarrow$  m ans)
```

so that a computation is represented by a function from a *continuation* for its result (a monadic function into an answer type) to the computation of the answer. In the world of arrows, we can represent a continuation by an arrow, rather than a function, and a continuation-passing arrow from b to c as a function from the continuation of the result to the continuation of the argument:

newtype *CPSFunctor* *ans* $a\ b\ c = \text{CPS } ((a\ c\ \text{ans}) \rightarrow (a\ b\ \text{ans}))$

Lifting an arrow to the CPS type is straightforward:

$\text{liftCPS} :: \text{Arrow } a \Rightarrow a\ b\ c \rightarrow \text{CPSFunctor } \text{ans } a\ b\ c$
 $\text{liftCPS } f = \text{CPS } (\lambda k \rightarrow f \ggg k)$

But now, in order to define the basic arrow operations on CPS arrows, we find we already need to use application at the underlying arrow type!

instance *ArrowApply* $a \Rightarrow \text{Arrow } (\text{CPSFunctor } \text{ans } a)$ **where**
 $\text{arr } f = \text{liftCPS } (\text{arr } f)$
 $\text{CPS } f \ggg \text{CPS } g = \text{CPS } (\lambda k \rightarrow f\ (g\ k))$
 $\text{first } (\text{CPS } f) = \text{CPS } (\lambda k \rightarrow \lambda \bullet (b, d) \rightarrow f\ (\lambda \bullet c \rightarrow k\ \bullet (c, d)) \bullet b)$

To define $\text{first } (\text{CPS } f)$ we must invoke f with a continuation which recombines its result with the second component of the argument. This we can do, but only in the scope of $\lambda \bullet (b, d) \rightarrow \dots$ which binds a name to that second component. We can only construct the arrow representing f 's continuation within another arrow, and so we can only construct the continuation of f 's argument within an arrow, which forces us to use *app* to invoke it. In the code above, we see that we require an arrow in class *ArrowApply* since the arrow-bound variable d appears in an expression to the left of an arrow application. In a way, since continuation passing is the epitome of higher-order programming, this is not really surprising.

CPS arrows inherit the ability to support failures and failure handling from the underlying arrow type, and can of course support dynamic choice and application. However, we will not give the definition here. What we will do is show how to define a *jump* operator, which invokes a continuation supplied as its input

$\text{jump} :: \text{ArrowApply } a \Rightarrow \text{CPSFunctor } \text{ans } a\ (a\ c\ \text{ans}, c)\ z$
 $\text{jump} = \text{CPS } (\lambda k \rightarrow \text{app})$

and a combinator *calcc*, which passes the current continuation to its argument arrow:

$\text{calcc} :: \text{ArrowApply } a \Rightarrow$
 $(a\ c\ \text{ans} \rightarrow \text{CPSFunctor } \text{ans } a\ b\ c) \rightarrow \text{CPSFunctor } \text{ans } a\ b\ c$
 $\text{calcc } f = \text{CPS } (\lambda k \rightarrow \text{let } \text{CPS } g = f\ k \text{ in } g\ k)$

As we have seen, continuation passing arrows always support application, and must be based on an underlying arrow type which also supports application. Thus, both the argument and the resulting arrow types correspond to monads. Our CPS functor

is therefore no more general than the CPS monad transformer, but nonetheless, what we have shown is that we can work entirely with arrows even if we want to use continuation passing style.

10. Active web pages: CGI programs as arrows

So far in this paper we have shown how the arrow interface can generalise a variety of existing combinator libraries. In this section we shall discuss a library we are currently developing, which was inspired by the concept of arrows.

The application that this library addresses is that of constructing active web pages, that is, pages that may appear differently each time they are visited. Active web pages are represented by programs, which may run either in the client browser (applets) or on the web server. Quite different technologies are used in each case; we concern ourselves here with programs which run on a web server. Such programs can query a database held on the server, allow clients to upload new data, and so on. Even rather simple programs can be very useful: for example, those which enable students to book meetings with a teacher, or researchers to submit articles to conferences.

Active web pages of this sort are implemented by so-called *CGI programs* stored on the server. When a client accesses the URL of the program, then it is run on the server, and the output from the program (usually HTML) is sent back to the client browser. There are a couple of different mechanisms for sending data from the client to the CGI program; the one we will consider sends an encoding of the fields of an HTML form to the web server, along with the request to run the program. CGI stands for *Common Gateway Interface*, the protocol governing the form in which data is sent to and fro between the client and the server.

Unfortunately, this mechanism is awkward to use in practice. Normally, the implementor of a CGI program wishes to lead the remote client through a series of interactions, for example, first asking a student to identify him or herself, then offering a choice of meeting times, then confirming that a time has been booked. But interactions with the client can only take place in between runs of CGI programs. To ask the client a question, a CGI program must output the question as an HTML form, and terminate. When the client answers the question by filling in and submitting the form, then in general a different CGI program is run to accept and process the answer. This leads to poor modularity, because the format of the form (field names, etc.) must be known both to the program which creates it, and to the program which interprets its contents. But a more severe problem is that the *state* of the CGI program is lost across the interaction.

It is therefore necessary to save the state of the CGI program *explicitly* across each interaction. *This cannot conveniently be done on the server!* It is by no means certain that the client ever will submit a reply, so that if the state were saved on the server then it might remain there for ever, waiting for a reply that never came. On the other hand, the client might submit a reply, then use the ‘Back’ button in the browser, and

reply to the same question again! If second and subsequent replies are to be handled properly, then the state cannot be discarded even once a reply has been received.⁵

The solution is to store the state of the CGI program *on the client*, along with the question. When the client submits an answer, then the state is returned along with it, permitting the CGI program to pick up from the same point that it left off. One can think of this state as a kind of continuation: when a CGI program wishes to ask the client something, it captures its current continuation and sends it along with the question to the client, and when the client replies then the continuation is returned to the server, and can be invoked to handle the reply. HTML provides a mechanism for handling such data: an HTML form can contain ‘hidden fields’ whose contents are returned unchanged to the server when the form is filled in and submitted. Unfortunately, though, HTML fields cannot contain function values, and so we must find a different way to represent continuations if we are to use this idea.

The combinator library I am developing takes care of suspension of computations, saving of state, and restart from the same point. It lets the CGI programmer view interaction with the client as a procedure call; there is an arrow

ask :: CGI String String

which maps a question to the client’s answer. Thus programs which conduct a series of interactions can be implemented very simply. For example, consider a program which asks the client

What is your question?

expecting to receive a reply such as

How old are you?

and then asks the client the same question, taking the client’s answer, and then finally sending the client a result such as

The answer to “How old are you?” is 40.

Such a program can be implemented by the arrow

```
arr (λz → “What is your question?”) >>> ask >>>
(arr id &&& ask) >>>
arr (λ(q,ans) → “The answer to\”++q++“\” is” ++ans)
```

or, using Paterson’s syntax,

```
λ • z → q ← ask • “What is your question?”
ans ← ask • q
“The answer to\”++q++“\” is” ++ans
```

⁵ It would be possible to save the state on the server along with an appropriate key to enable the server to recognise the same client returning in the same state, given a policy for deletion of saved states, but this is a more complicated solution than the one we propose.

Why choose the arrow interface rather than the monad interface for this problem? The key observation guiding the choice was that the combinators need to save the entire state of the program at an *ask* operation. But the state of a functional program is the collection of values bound to variables, and there is no way for any Haskell function to capture the values of variables in scope at the point where it is called. Since we cannot capture the values of variables when *ask* is invoked, we must see to it that there *are* no variables in scope at these points. Fortunately, we need only be concerned here with variables bound to the results of computations, since it is only these that may take a different value the next time the program is run. Now we see that monadic bind is problematic: it allows us to bind a local variable to the result of a computation, with a scope which extends over another computation, and potentially an *ask*. But the arrow interface does not permit this: the only way to bind a variable to the result of a computation is with the *arr* combinator. Using *ask* within an argument to *arr* constructs an arrow within an arrow, which is harmless unless the arrow is eventually invoked. The only way to do that is via *app*. Thus, provided we do not provide *app*, then we can guarantee that no variables bound to the result of a computation scope over an arrow (or at least, no arrow that is ever invoked).

Note that Paterson's syntax disguises this property a little: in the example above, *q* does appear to scope over a call of *ask*, but of course this is no longer true after desugaring. Provided no arrow-bound variables appear to the left of a \bullet , then Paterson's translation generates a term not involving *app*, in which no arrow-bound variables are in scope over another arrow.

How, then, can CGI arrows be represented? When such an arrow is invoked, it may either terminate normally, producing a result, or it may suspend at an *ask* operation. On suspension, an arrow must produce a state to save, and a question to ask. A CGI arrow can also be entered in two different ways: it may either be entered normally, with an argument, or it may be resumed from an *ask*. In the latter case we must supply a state to resume from, and the answer to the question. A natural representation for CGI arrows might therefore be

```
newtype CGI0 b c =
  CGI0 (Either b (ScriptState,String) → Either c (ScriptState,String))
```

However, in general a CGI program may have side-effects on the server, which this type does not allow for. So we shall instead represent CGI arrows as *arrows* between these two types, which in practice will be arrows which can perform I/O. We shall parameterise our definitions on the underlying arrow type, and so define a CGI functor:

```
newtype CGIFunctor a b c =
  CGI (a (Either b (ScriptState,String)) (Either c (ScriptState,String)))
```

Now we can define

type *CGI* *b c* = *CGIFunctor* (*Kleisli IO*) *b c*

With this definition, the *ask* operation is easily defined: it suspends when entered normally, and delivers the answer as its result when it is resumed. No state is needed to resume the *ask* operator itself, so we assume that the *ScriptState* type includes a constructor *Empty*:

data *ScriptState* = *Empty* | ...

We define *ask* (using Paterson’s notation) as follows:

ask :: *ArrowChoice a* ⇒ *CGIFunctor a String String*
ask = *CGI* ($\lambda \bullet z \rightarrow$ **case** *z* **of**
 Left q \rightarrow *Right (Empty, q)*
 Right (Empty, a) \rightarrow *Left a*)

The first alternative here handles a normal entry, and suspends to ask the question *q*, while the second alternative handles a resumption, and delivers the answer *a* as the arrow’s result.⁶

The *arr* operator is also easily defined: a pure arrow can never suspend, and therefore can never be resumed either, so we need to consider only the *Left* summands here.

instance *ArrowChoice a* ⇒ *Arrow (CGIFunctor a)* **where**
 arr f = *CGI* ($\lambda \bullet (Left\ b) \rightarrow Left\ (f\ b))$
 :
 :

It is when we define arrow composition that we first need to make use of the state. A composition of arrows may suspend either in the first arrow, or in the second, and the state that we save must record which case applied. Similarly, when we resume a composition of arrows, then we need to know which arrow to resume. We shall therefore extend the *ScriptState* type to record this information:

data *ScriptState* = *Empty* | *InLeft ScriptState* | *InRight ScriptState* | ...

⁶ Note that here and elsewhere, we assume that the state returned from the client matches the state produced when the CGI program suspends. There is an opportunity here for a malicious client to fabricate his own state, and thus induce the program to perform actions its author did not intend. Encrypting the state before sending it to the client might be a way to solve this problem.

The definition of composition then becomes

```

instance ArrowChoice a  $\Rightarrow$  Arrow (CGIFunctor a) where
  ⋮
  CGI f  $\ggg$  CGI g =
    CGI ( $\lambda z \bullet z \rightarrow$  case z of
      Left b            $\rightarrow$  enterf  $\bullet$  Left b
      Right (InLeft s, a)  $\rightarrow$  enterf  $\bullet$  Right (s, a)
      Right (InRight s, a)  $\rightarrow$  enterg  $\bullet$  Right (s, a))

  where enterf  $\bullet$  x = y  $\leftarrow$  f  $\bullet$  x
           case y of
             Left c        $\rightarrow$  enterg  $\bullet$  Left c
             Right (s, q)  $\rightarrow$  Right (InLeft s, q)

           enterg  $\bullet$  x = y  $\leftarrow$  g  $\bullet$  x
           case y of
             Left d        $\rightarrow$  Left d
             Right (s, q)  $\rightarrow$  Right (InRight s, q)
  ⋮

```

The first case in \ggg handles initial entry to the composition, and just makes an initial entry to f . The second and third cases handle resumption of f and g , respectively. Arrow $enterf$ invokes f , and if f terminates normally, makes an initial entry to g . If f suspends, on the other hand, then $enterf$ records that the suspension occurred in the left operand of \ggg . Arrow $enterg$ similarly records that a suspension in g occurred in the right operand of \ggg . Thus, we always record in which arrow a suspension occurred, and on resumption we return to the same point.

When we define the *first* combinator, we need to use the state in a different way. There is no need to record where a suspension occurred: when *first* f suspends, it must be in the arrow f . However, since *first* f must preserve the second component of its input unchanged, then when we resume after a suspension, we need to know what the value of this second component was. We therefore have to save it in the state. One difficulty is that the values to be saved can have many different types, at different occurrences of *first*. We shall convert them all to the *same* type before saving them; since states must eventually be embedded in HTML fields, it is convenient to convert them to strings, using Haskell's standard function *show*. When we resume from such a state, we can convert the saved value back to its original type using the standard function *read*, which satisfies $read \circ show = id$.

We shall therefore extend the *ScriptState* type again:

```

data ScriptState = Empty |
                  InLeft ScriptState | InRight ScriptState |
                  Save String ScriptState

```


and define *first* as

```

instance ArrowChoice  $a \Rightarrow$  Arrow (CGIFunctor  $a$ ) where
  ⋮
  — first :: (Show  $d$ , Read  $d$ )  $\Rightarrow$ 
      CGIFunctor  $a$   $b$   $c \rightarrow$  CGIFunctor  $a$  ( $b, d$ ) ( $c, d$ )

  first (CGI  $f$ ) = CGI ( $\lambda \bullet x \rightarrow$ 
    case  $x$  of
      Left ( $b, d$ )       $\rightarrow$  enterf  $\bullet$  (Left  $b, d$ )
      Right (Save  $v$   $s, a$ )  $\rightarrow$  enterf  $\bullet$  (Right ( $s, a$ ), read  $v$ ))

  where enterf  $\bullet$  ( $x, d$ ) =  $y \leftarrow f \bullet x$ 
    case  $y$  of
      Left  $c$        $\rightarrow$  Left ( $c, d$ )
      Right ( $s, q$ )  $\rightarrow$  Right (Save (show  $d$ )  $s, q$ )

```

On an initial entry to *first* f , we just invoke f directly; on a resumption we reconstruct the saved second component from the state before doing so. On final termination of f , we just pair its output with the second component d , but on suspension we save d in the state. Note that the type d must support *read* and *show* operations, which not all types do. This is recorded in the type signature of *first*, which requires d to be an instance of the classes *Read* and *Show*.

CGI arrows also permit dynamic choices. Implementing *left* turns out to be particularly simple, because *left* f can suspend only if the input was of the form *Left* b ; we therefore don't need to record any additional information in the state to allow us to decide whether or not to invoke f on a resumption.

```

instance ArrowChoice  $a \Rightarrow$  ArrowChoice (CGIFunctor  $a$ ) where
  left (CGI  $f$ ) = CGI ( $\lambda \bullet x \rightarrow$ 
    case  $x$  of
      Left (Left  $b$ )       $\rightarrow$  enterf  $\bullet$  Left  $b$ 
      Left (Right  $d$ )      $\rightarrow$  Left (Right  $d$ )
      Right ( $s, a$ )       $\rightarrow$  enterf  $\bullet$  Right ( $s, a$ ))

  where enterf  $\bullet$   $x$  =  $y \leftarrow f \bullet x$ 
    case  $y$  of
      Left  $c$        $\rightarrow$  Left (Left  $c$ )
      Right ( $s, q$ )  $\rightarrow$  Right ( $s, q$ )

```

When *left* f is invoked normally, we either invoke f normally (on arguments tagged *Left*), or return arguments tagged *Right* directly. On a resumption of *left* f , we just resume f in the same state. The arrow *enterf* invokes f : when it terminates normally,

then *enterf* also does so (tagging the result *Left*). When *f* suspends then so does *enterf*, in the same state.

It is also possible to give an appealing interpretation of *zeroArrow* and $\langle + \rangle$ for CGI arrows: $f \langle + \rangle g$ creates two ‘threads’ which run ‘in parallel’, and *zeroArrow* terminates a thread. (These ‘threads’ are simulated in ordinary Haskell, no concurrency is required in the underlying implementation). We use this mechanism to enable a CGI arrow to ask several questions in one interaction (if both *f* and *g* suspend). We omit the details here.

It is not possible, however, to implement *app*. The difficulty here is that the types that CGI arrows operate over must support *read* and *show*, so that intermediate values can be saved on the client. CGI arrows themselves are implemented in terms of functions, and so cannot be read and written. Therefore, a CGI arrow cannot take another CGI arrow in its input, and *app* cannot be defined.

The library I am developing is based on the ideas in this section, but is necessarily a little more complicated. It is an oversimplification to consider the communication with the client to consist of a single question and answer, or even multiple questions and answers. In reality the client is sent an HTML page containing one or more HTML forms, each of which may contain many fields. The full-scale library includes combinators for generating various HTML elements, and for putting parts of forms together into larger forms. There is also a ‘top-level’ function

$$\text{serveCGI} :: \text{CGI } a \ b \rightarrow a \rightarrow \text{IO } b$$

which takes an arrow and ‘runs it’, taking care of encoding states in hidden fields, decoding the data returning from the client, and so on. We omit the definition here, since it is complicated by the vagaries of HTML and the CGI protocol.

One major irritation which we have so far glossed over is that CGI arrows cannot actually be made an instance of the *Arrow* class defined in this paper! In a sense, we have defined *CGIFunctor* to be ‘the wrong type’, but Haskell does not allow us to define the right one! An extension to the Haskell type system would be needed to make CGI arrows into arrows as we defined them earlier.

The problem lies in the types of the arrow methods given in this section. Look back at the type of *first*: it requires that the type of the value to be saved be an instance of the *Read* and *Show* classes. The type given for *first* in the definition of the *Arrow* class makes no such restriction. Therefore this implementation of *first* cannot be declared to be an instance of the generic one – it is less general.

We might attempt to solve this problem by moving the restriction to a different place. Let us redefine the CGI arrow type so that it is only applicable to types in these classes:

$$\begin{aligned} \text{newtype } (\text{Read } b, \text{Show } b, \text{Read } c, \text{Show } c) \Rightarrow \\ \text{CGIFunctor } a \ b \ c = \\ \text{CGI } (a \ (\text{Either } b \ (\text{ScriptState}, \text{String})) \ (\text{Either } c \ (\text{ScriptState}, \text{String}))) \end{aligned}$$

In categorical terms, we want to define a new category whose arrows are CGI arrows, and whose objects are a *subset* of the Haskell types, namely those supporting *read* and

show. Now, since the implementation of *first* given in this section constructs a CGI arrow from (b, d) to (c, d) , then it is evident that the type d must support *read* and *show*, and there is no need to explicitly require that in the type of *first*. As a result, it should now be possible to declare CGI arrows an instance of the generic arrow class.

Unfortunately, this does not work. The Haskell type system requires the restrictions on d in the type of *first*, even if we declare that they are satisfied for all CGI arrows. Haskell does not infer from the occurrence of a type *CGI* $b\ c$, that b and c must be instances of *Read* and *Show* – and indeed, this is not even true, because of the way that type restrictions on datatype definitions are interpreted. I consider this to be a defect of the Haskell type system, which hopefully can be corrected in a future version of the language.

In the absence of such a correction, we are obliged to make a copy of the arrow library, and all the generic code that uses it, with the only difference that the type assigned to *first* in the *Arrow* class is the one required for the CGI instance. By doing so we can still benefit from using a standard arrow interface to the CGI library – we can still combine CGI arrows with other arrow code – but any program which uses the CGI library must import a special definition of the arrow class, which restricts *all* arrows in the entire program to work over types supporting *read* and *show*. This is frustrating indeed.

Finally, we note with hindsight that a monadic interface *could* be used instead here. We could define a monad whose computations can be suspended and resumed, in an analogous way to CGI arrows. However, the definition of $m \gg= f$ would need to record not only which of m or f suspended, but also the *value* that m delivered, if suspension occurred in f . Concretely, the ‘*InRight*’ form of *ScriptState* would need to carry an extra component, namely the value of m . Thus the problem of recording free variables is solved: every free variable of an *ask* operation which is bound to the result of a computation, is bound by an occurrence of $\gg=$, and we can make that occurrence of $\gg=$ responsible for saving the value.

However, even if a monadic interface would be possible, we believe it would make for less efficient CGI programs. The monadic library we suggest would need to save *every* previously delivered value, whereas the arrow library saves only those which are still needed. Thus the monadic library would tend to send more information to and from the client. Of course, such a monadic library would also fall foul of the typing problem just discussed, so that a CGI monad could not be declared to be an instance of Haskell’s *Monad* class. Consequently it could not be used together with standard monadic functions, or Haskell’s monadic **do** syntax.

11. Conclusions

This paper proposes the replacement of monads as a structuring tool for combinator libraries, by arrows. We have seen that any monadic library can be given an arrow interface instead (via *Kleisli* arrows), and so the arrow interface is strictly more general.

We have seen that many monadic programming techniques have analogues in the world of arrows: monad transformers become functors, standard monad constructions for exceptions, state passing and continuations carry over to arrows, even generic monadic functions often have an arrow analogue. But basing an interface on arrows instead of monads permits finer distinctions to be made: we can distinguish between kinds of computation which permit dynamic choices to be made, or dynamic computations to be invoked, and those which do not.

The advantage of the arrow interface is that it has a wider class of implementations than the monad interface does; it is more general. Thus, some libraries based on abstract data types which simply are not monads, can nonetheless be given an arrow interface. Such libraries include those for processes modelled by stream processors or fudgets, libraries for efficient parsing, or in general any library which computes static properties of computations in advance of running them. So this category includes a number of libraries which are highly useful in practice. By giving them an arrow interface, we make it possible to use them together with generic arrow code.

Moreover, some existing *monadic* libraries might benefit by replacing the monads with arrows. One motivation might be in order to introduce the same kind of optimisation which Swierstra and Duponcheel used. We believe this may be the case for Conal Elliot's animation library [4], and for Bjesse et al. library for hardware design [1].

On the negative side, arrows oblige one to program in a 'point-free' style which can be unnatural, unless Paterson's syntactic sugar is available. Moreover, while the designer of a monad based library need only implement *return* and \gg to define his monad, the designer of an arrow based library must implement at least *arr*, \gg and *first*, and possibly *left* and *app* as well. In cases when a library can be based on a monad, it might thus be sensible to build an arrow version on top of a monadic one using the Kleisli construction, to reduce the programming effort required. Finally, arrow programs will require heavy optimisation if they are to perform well, because of the large number of calls to very small functions.

Notwithstanding these disadvantages, we believe that arrows offer a useful extension to the generality of generic library interfaces.

Acknowledgements

I am grateful to Ross Paterson, Phil Wadler, and the anonymous referees for their very careful comments on this paper. Lennart Augustsson helped clarify the ideas herein at an earlier stage. Finally, thank you to Johan Jeuring for his patient editing.

References

- [1] P. Bjesse, K. Claessen, M. Sheeran, S. Singh, Lava: hardware design in haskell, in: Internat. Conf. on Functional Programming, Baltimore, ACM, New York, 1998.
- [2] M. Barr, C. Wells, Category Theory for Computing Science, Prentice-Hall, Englewood Cliffs, NJ, 1990.

- [3] M. Carlsson, T. Hallgren, FUDGETS - a graphical user interface in a lazy functional language, in: FPCA '93 – Conf. on Functional Programming Languages and Computer Architecture, ACM Press, New York, June 1993, pp. 321–330.
- [4] C. Elliott, P. Hudak, Functional reactive animation, in: Internat. Conf. on Functional Programming, ACM SIGPLAN, New York, 1997.
- [5] P. Hudak et al., Report on the Programming Language Haskell: A Non-Strict, Purely Functional Language, March 1992. Version 1.2. Also in Sigplan Notices, May 1992.
- [6] D. King, P. Wadler, Combining monads, in: Glasgow Workshop on Functional Programming, Ayr, Springer, Berlin, July 1992.
- [7] S. Liang, P. Hudak, M. Jones, Monad transformers and modular interpreters, in: Conf. Record of POPL'95: 22nd ACM SIGPLAN-SIGACT Sympos. on Principles of Programming Languages, San Francisco, CA, January 1995, pp. 333–343.
- [8] S. MacLane, Categories for the Working Mathematician, Springer, Berlin, 1971.
- [9] J. Peterson, K. Hammond, The Haskell 1.3 Report, Technical Report YALEU/DCS/RR-1106, Yale University, New Haven, CT, 1996.
- [10] B.C. Pierce, Basic Category Theory for Computer Scientists, The MIT Press, Cambridge, MA, 1991.
- [11] S.D. Swierstra, L. Duponcheel, Deterministic, error-correcting combinator parsers, in: J. Launchbury, E. Meijer, T. Sheard (Eds.), Advanced Functional Programming, Lecture Notes in Computer Science – Tutorial, Vol. 1129, Springer, Berlin, 1996, pp. 184–207.
- [12] P. Wadler, How to replace failure by a list of successes, in: Proc. 1985 Conf. on Functional Programming Languages and Computer Architecture, Nancy, France, 1985, pp. 113–128.
- [13] P. Wadler, Comprehending Monads, in: Proc. 1990 ACM Conf. on Lisp and Functional Programming, Nice, France, 1990, pp. 61–77.
- [14] P. Wadler, The essence of functional programming, in: Proc. 1992 Symp. on Principles of Programming Languages, Albuquerque, New Mexico, 1992, pp. 1–14.
- [15] P. Wadler, Monads for functional programming, in: J. Jeuring, E. Meijer (Eds.), Advanced Functional Programming, Lecture Notes in Computer Science, Vol. 925, Springer, Berlin, May 1995, pp. 24–52.